iHistorian® Collector Swapper utility
A command line utility to reconfigure all tags assigned to collector1 and assign them to collector2.
syntax:

CollectorSwap.exe SourceCollector TargetCollector
Where the SourceCollector is the collector to change tags from and TargetCollector is the collector to change tags to.

This was designed as a command line utility that could be called from the VB ‘Shell()’ function. Since large tag counts may take long times to write changes back to the historian, this will allow the VBA thread to make this call and return immediately, preventing a lockup of an HMI while tags were being processed.

A log file named CollectorSwap_YYYYMMDD.LOG will be created in the utility’s folder each time it is run. Log entries will be appended until the next day, when a new log is started.

Using the collector status tags with the iFIX® collector, an event can be generated in the iFIX® scheduler that can call the shell function to swap tags from a primary to a secondary collector.

This would be a stop-gap until iHistorian® 3.0, when the collector redundancy feature will be built-in to the product.

On my Centrino laptop, 5000 tags takes a few seconds to reassign. 20K tags take about a minute to reassign.

Core code:

   'LogCommand "CCollectorSwap.SwapCollectors: Entering"

    If Len(SourceCollector) = 0 Or Len(TargetCollector) = 0 Then

        SwapCollectors = False

        Exit Function

    End If

    '

    LogCommand "CCollectorSwap.SwapCollectors: Source : " & SourceCollector

    LogCommand "CCollectorSwap.SwapCollectors: Target : " & TargetCollector

    '

    Dim blnRetVal As Boolean

    Dim rstTags As iHistorian_SDK.TagRecordset

    Dim tagMaster As iHistorian_SDK.Tag

    '

    ConnectServer

    If Not m_oServer Is Nothing Then

        If m_oServer.Connected Then

            'Get A New Tag recordset

            Set rstTags = m_oServer.Tags.NewRecordset

            With rstTags

                With .Fields

                    .Clear

                    .Tagname = True

                    .CollectorName = True

                End With

                'Fill In Criteria, Get All Tags for the SourceCollector

                With .Criteria

                    .Tagname = "*"

                    .CollectorName = SourceCollector

                End With

                ' perform query

                .QueryRecordset

                ' did we get any??

                If .Item.Count > 0 Then

                    LogCommand "CCollectorSwap.SwapCollectors: Found " & .Item.Count & " tags for [" & SourceCollector & "]"

                    ' get master record

                    Set tagMaster = .Master

                    ' change collector from source to target

                    tagMaster.CollectorName = TargetCollector

                    'Select All Tags In The recordset

                    .SelectAll

                    'Commit Changes

                    If .WriteRecordset(True) Then

                        blnRetVal = True

                        LogCommand "CCollectorSwap.SwapCollectors: WriteRecordset successful for [" & TargetCollector & "]"

                        '

                        ' restart the collector to make the changes go into affect

                        LogCommand "CCollectorSwap.SwapCollectors: Restarting [" & TargetCollector & "]"

                        m_oServer.Collectors.Item.Item(TargetCollector).Running = False

                        m_oServer.Collectors.Item.Item(TargetCollector).Running = True

                    Else

                        LogCommand "CCollectorSwap.SwapCollectors: WriteRecordset failed for [" & TargetCollector & "] [" & .LastError & "]"

                    End If

                Else

                    blnRetVal = True

                    LogCommand "CCollectorSwap.SwapCollectors: No tags found for [" & SourceCollector & "]"

                End If

            End With

        Else

            LogCommand "CCollectorSwap.SwapCollectors: Unable to connect to default server [" & m_strDefaultServer & "]"

        End If

    Else

        LogCommand "CCollectorSwap.SwapCollectors: default server not initialized [" & m_strDefaultServer & "]"

    End If

PROC_EXIT:

    On Error Resume Next

    'LogCommand "CCollectorSwap.SwapCollectors: Exiting"

    Set tagMaster = Nothing

    Set rstTags = Nothing

    '
In order to make this solution work, you need to monitor the status of the two collectors that are to be responsible for the tags. We will call them ‘CollectorA’ and ‘CollectorB’.

There are a few ways to monitor collector status. One is with the ‘Collector Status Output’ tags. These are the ‘Rate’, ‘Status’, and ‘Heartbeat’ feedback tags that are configured in the ‘Advanced’ tab of the iHistorian® Collector configuration screens. This would require you to set up tags in the iFIX® process database (for the iFIX®/FIX32® collectors) or in the PLC (if using the OPC collector) to check the values of these tags and trigger an event if the status indicated a collector failure. For instance, the Hearbeat tag will set the value of a DI/DO/DR to a ‘1’ every minute. You would need a watchdog proram block to watch for this, and reset it to zero. If the collector does not set it back to one after a minute, you know the collector is not running. The ‘Status’ feedback tag will write a string to a text block with the current value of the collector status (Unknown, Starting, Running, Stopping, Stopped). If the value of this text block is not ‘Starting’ or ‘Running’, again you know the collector is down.

Alternatively, you can use the iFIX® Scheduler to check the status of the collector. The easiest method, without using tags, is an ADO recordset that periodically checks the ‘Status’ field of the collectors. Just like with the ‘Status’ feedback tag above, if the value of the Status field is not ‘Starting’ or ‘Running’, the collector is down.
Once you have determined that the collector is down using one of the methods mentioned above, you then need to tell iHistorian® to move the tags for that collector to the backup collector. This is where the CollectorSwap utility comes in. Simply use ‘Runtask’ (available with documentation from the http://support.gefanucautomation.com downloads section) to call the command line documented above.

You can also continue monitoring the collector status, and when the primary collector comes back on, call the CollectorSwap utility again, and change the order so that the tags are reassigned to the primary collector when it is running.

Sample VBA code for the included iFIX® 3.5 schedule:

'---------------------------------------------------------------------------------------

' Module    : ihCollectorStatus

' DateTime  : 6/11/2004 13:45

' Author    : nhenderson

' Purpose   : demonstrate using ADO to monitor collector status

'

' this project requires that the OLE-DB provider for iHistorian be installed on this node

' make sure a reference to the Microsoft ActiveX Data Objects library is set

' in the Tools->References menu.

' this sample is using MSADO 2.6

'---------------------------------------------------------------------------------------

'

' create ADO objects once for project, instead of each time it fires

' this is more efficient, and uses less resources than creating and destroying a recordset

' each time the event fires

Private m_rstCollectorStatus As ADODB.Recordset

Private m_cnnCollectorConnection As ADODB.Connection

'

'---------------------------------------------------------------------------------------

' Procedure : CFixScheduler_Initialize

' DateTime  : 6/11/2004 13:46

' Author    : nhenderson

' Purpose   : fires on run mode start

'---------------------------------------------------------------------------------------

Private Sub CFixScheduler_Initialize()

    On Error GoTo ERR_HANDLER

    Set m_rstCollectorStatus = New ADODB.Recordset

    Set m_cnnCollectorConnection = New ADODB.Connection

    '

    ' is security is enabled, update with your username and password

    ' secure this vba project with a password, so that the info is

    ' not available for easy access

    m_cnnCollectorConnection.Open "Provider=ihOLEDB.iHistorian.1;User Id=;Password="

    ' sanity checks

    If m_cnnCollectorConnection.State = ADODB.adStateOpen Then

        m_rstCollectorStatus.Open "SELECT CollectorName, Status FROM ihCollectors", m_cnnCollectorConnection, adOpenStatic

    End If

PROC_EXIT:

    On Error GoTo 0

    Exit Sub

ERR_HANDLER:

    'MsgBox "Error " & Err.Number & " (" & Err.Description & ") in procedure CFixScheduler_Initialize of VBA Document ihCollectorStatus"

    Debug.Print "Project_ihCollectorStatus.ihCollectorStatus.CFixScheduler_Initialize Error: [" & Err.Number & "] " & Err.Description

    'Resume Next

    Resume PROC_EXIT

End Sub

'

'---------------------------------------------------------------------------------------

' Procedure : CFixScheduler_InitializeConfigure

' DateTime  : 6/11/2004 13:46

' Author    : nhenderson

' Purpose   : fires when going to configure mode

'---------------------------------------------------------------------------------------

Private Sub CFixScheduler_InitializeConfigure()

    On Error Resume Next

PROC_EXIT:

    On Error GoTo 0

    ' cleanup

    If Not m_rstCollectorStatus Is Nothing Then

        If m_rstCollectorStatus.State = ADODB.adStateOpen Then m_rstCollectorStatus.Close

        Set m_rstCollectorStatus = Nothing

    End If

    If Not m_cnnCollectorConnection Is Nothing Then

        If m_cnnCollectorConnection.State = ADODB.adStateOpen Then m_cnnCollectorConnection.Close

        Set m_cnnCollectorConnection = Nothing

    End If

    Exit Sub

ERR_HANDLER:

    'MsgBox "Error " & Err.Number & " (" & Err.Description & ") in procedure CFixScheduler_InitializeConfigure of VBA Document ihCollectorStatus"

    Debug.Print "Project_ihCollectorStatus.ihCollectorStatus.CFixScheduler_InitializeConfigure Error: [" & Err.Number & "] " & Err.Description

    'Resume Next

    Resume PROC_EXIT

End Sub

'---------------------------------------------------------------------------------------

' Procedure : ihCollectorMonitor_OnTimeOut

' DateTime  : 6/11/2004 13:47

' Author    : nhenderson

' Purpose   : fires at event interval

'---------------------------------------------------------------------------------------

Private Sub ihCollectorMonitor_OnTimeOut(ByVal lTimerId As Long)

    On Error GoTo ERR_HANDLER

    If m_rstCollectorStatus.State = ADODB.adStateOpen Then m_rstCollectorStatus.Requery

    Do While Not m_rstCollectorStatus.EOF

        ' uncomment this line to see what is being detected

        'Debug.Print m_rstCollectorStatus.Fields("CollectorName").Value & " : " & m_rstCollectorStatus.Fields("Status").Value

        Select Case m_rstCollectorStatus.Fields("Status").Value

        Case "Starting", "Running"

            ' check here to see if the primary is back up and running

            ' if so, call the collector swap to return the tags back to the primary

            Select Case m_rstCollectorStatus.Fields("CollectorName").Value

            Case "CollectorA"

                Shell System.DocumentPath & "\CollectorSwap.exe " & "CollectorB CollectorA", vbHide

            Case "CollectorB"

            End Select

        Case "Uknown", "Stopping", "Stopped"

            ' check for collector failure

            ' swap tags to the secondary

            Select Case m_rstCollectorStatus.Fields("CollectorName").Value

            Case "CollectorA"

                Shell System.DocumentPath & "\CollectorSwap.exe " & "CollectorA CollectorB", vbHide

            Case "CollectorB"

                Shell System.DocumentPath & "\CollectorSwap.exe " & "CollectorB CollectorA", vbHide

            End Select

        End Select

        ' look at next collector

        m_rstCollectorStatus.MoveNext

    Loop

PROC_EXIT:

    On Error GoTo 0

    Exit Sub

ERR_HANDLER:

    'MsgBox "Error " & Err.Number & " (" & Err.Description & ") in procedure ihCollectorMonitor_OnTimeOut of VBA Document ihCollectorStatus"

    Debug.Print "Project_ihCollectorStatus.ihCollectorStatus.ihCollectorMonitor_OnTimeOut Error: [" & Err.Number & "] " & Err.Description

    'Resume Next

    Resume PROC_EXIT

End Sub

Pertinent KnowledgeBase articles:

i020636 - iHistorian With Redundant iFIX Scada Nodes
http://globalcare.gefanuc.com/kc/kb/kbinfo.asp?objid=782310&id=268439413
i020600 - Redundacy and iHistorian Servers

http://globalcare.gefanuc.com/kc/kb/kbinfo.asp?objid=782310&id=268439377
i020667 - addressing redundancy with SCADA configurations

http://globalcare.gefanuc.com/kc/kb/kbinfo.asp?objid=782314&id=268439444
i020707 - Is there a heartbeat or watchdog between the collector and the server even when there are no tags to collect?

http://globalcare.gefanuc.com/kc/kb/kbinfo.asp?objid=782314&id=268439484
i020303 - Using a program block and a collector heartbeat to restart a stopped iHistorian Collector

http://globalcare.gefanuc.com/kc/kb/kbinfo.asp?objid=782317&id=268439080
i022081 - How would I setup a VISICONX object that connects to iHist?

http://globalcare.gefanuc.com/kc/kb/kbinfo.asp?objid=782317&id=268448712
i014538 - Copying iHistorian data from one tag to another tag

http://globalcare.gefanuc.com/kc/kb/kbinfo.asp?objid=782317&id=268438750
iHcopydata
http://globalcare.gefanuc.com/devsupport/showtools.asp?objid=198
i020945 - iFIX and FIX32 Collectors To Recognize SCADA Redundancy Scheme

http://globalcare.gefanuc.com/kc/kb/kbinfo.asp?objid=782314&id=268445285
